ganzrationale Funktionenschar

```
> restart:with(plots):
```

Die Funktionenschar :

```
> f:=(x,t)->x^4-t^2*x^2;
```

$$f := (x, t) \rightarrow x^4 - t^2 x^2$$

Die Nullstellen:

```
> Nullstellen:=[solve(f(x,t)=0,x)];
```

$$Nullstellen := [0, 0, -t, t]$$

$x = 0$ ist doppelte Nullstelle - starker Hinweis auf Extremum

Die drei Ableitungen :

```
> f1:=unapply(diff(f(x,t),x),(x,t));
```

$$f1 := (x, t) \rightarrow 4\,x^3 - 2\,t^2\,x$$

```
> f2:=unapply(diff(f1(x,t),x),(x,t));
```

$$f2 := (x, t) \rightarrow 12\,x^2 - 2\,t^2$$

```
> f3:=unapply(diff(f2(x,t),x),(x,t));
```

$$f3 := (x, t) \rightarrow 24\,x$$

Die Nullstellen der 1. Ableitung :

```
> Kandidaten:=[solve(f1(x,t)=0,x)];
```

$$Kandidaten := \left[0, \frac{1}{2}\sqrt{2}\,t, -\frac{1}{2}\sqrt{2}\,t\right]$$

```
> f2(Kandidaten[1],t);
```

$$-2\,t^2$$

```
> f(Kandidaten[1],t);
```

$$0$$

Weil die 2. Ableitung an der Stelle $x = 0$ negativ ist, liegt bei

(0 / 0) ein Hochpunkt

```
> f2(Kandidaten[2],t);
```

$$4\,t^2$$

```
> f(Kandidaten[2],t);
```

$$-\frac{1}{4}t^4$$

Weil die 2. Ableitung positiv ist - und wegen der Achsensymmetrie
bzgl. der y-Achse liegt bei ($\frac{1}{2}\sqrt{2}\,t$ / $-\frac{1}{4}t^4$) und ($-\frac{1}{2}\sqrt{2}\,t$ / $-\frac{1}{4}t^4$)

jeweils ein Tiefpunkt.

Die Nullstellen der 2. Ableitung :

```
> wpk:=[solve(f2(x,t)=0,x)];
```

$$wpk := \left[\frac{1}{6}\sqrt{6}\,t, -\frac{1}{6}\sqrt{6}\,t\right]$$

```
> f3(wpk[1],t);
```

$$4\sqrt{6}\,t$$

Fallunterscheidung bezüglich t bringt keine neuen Erkenntnisse,
da für $|t|$ die gleichen Funktionsgraphen erzeugt werden.

```
> f(wpk[1],t);
```

$$-\frac{5}{36}t^4$$

Die Wendepunkte sind : ($\frac{1}{6}\sqrt{6}\,t$ / $-\frac{5}{36}t^4$) und ($-\frac{1}{6}\sqrt{6}\,t$ / $-\frac{5}{36}t^4$)

Die Fläche zwischen x-Achse und Funktionsgraph ist :

```
> Fläche:=abs(Int(f(x,t),x=-t..t))=abs(int(f(x,t),x=-t..t));
```

$$Fläche := \left|\int_{-t}^{t} x^4 - t^2 x^2\,dx\right| = \frac{4}{15}|t|^5$$

> **g:=x->-x^4;**

$$g := x \to -x^4$$

Der Funktionsgraph:

> **plot({f(x,1),f(x,2),f(x,3),g(x)},x=-4..4,y=-25..10);**



Die Ortskurve der Minima ist $\mathbf{g}(x) = -x^4$